# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=ad-ldap-enum
pkgver=88.60bc5bb
pkgrel=7
pkgdesc='An LDAP based Active Directory user and group enumeration tool.'
groups=('blackarch' 'blackarch-recon')
arch=('any')
url='https://github.com/CroweCybersecurity/ad-ldap-enum'
license=('MIT')
depends=('python' 'python-ldap3' 'python-openpyxl')
makedepends=('git' 'dos2unix')
source=("git+https://github.com/CroweCybersecurity/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

prepare() {
  cd $pkgname

  #sed -i 's|/env python|/env python2|' "$pkgname.py"

  dos2unix "$pkgname.py"
}

package() {
  cd $pkgname

  install -Dm 755 "$pkgname.py" "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md
  install -Dm 644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

